Image Cover Sheet

CLASSIFICATION	SYSTEM NUMBER 14540
UNCLASSIFIED	

TITLE

A COMPUTER PROGRAM FOR THE CALCULATION OF LIQUID OR VAPOUR PENETRATION THROUGH BARE AND CLOTHED SKIN

System Number:

Patron Number:

Requester:

Notes:

DSIS Use only:

Deliver to: JR

This page is left blank

This page is left blank

UNCLASSIFIED

72-03509

DRES

SUFFIELD TECHNICAL NOTE

NO. 310

A COMPUTER PROGRAM FOR THE CALCULATION OF LIQUID OR VAPOUR PENETRATION THROUGH BARE AND CLOTHED SKIN (U)

by

W.R. McPherson

PROJECT NO. 20-20-32

April 1972

DEFENCE RESEARCH ESTABLISHMENT SUFFIELD : RALSTON : ALBERTA

UNCLASSIFIED

SUFFIELD TECHNICAL NOTE NO. 310

A COMPUTER PROGRAM FOR THE CALCULATION OF LIQUID OR VAPOUR PENETRATION THROUGH BARE AND CLOTHED SKIN (U)

by

W.R. McPherson

PROJECT NO. 20-20-32

UNCLASSIFIED

DEFENCE RESEARCH ESTABLISHMENT SUFFIELD
RALSTON ALBERTA

SUFFIELD TECHNICAL NOTE NO. 310

A COMPUTER PROGRAM FOR THE CALCULATION OF LIQUID OR VAPOUR PENETRATION THROUGH BARE AND CLOTHED SKIN (U)

by

W.R. McPherson

ABSTRACT

A mathematical model has been developed by Monaghan at DRES to predict the penetration of liquid or vapour through bare or clothed skin; in order to calculate solutions to the equations of this model, a computer program was written. It is the objective of this paper to give a description of this computer program and provide the method of using it.

RÉSUMÉ

Un modèle mathématique a été developpé par Monaghan chez Centre de Recherches pour la Defense Suffield (CRDS) pour prédire la pénetration par un liquide ou une vapeur à travers la peau nue ou vêtue; pour calculer les solutions aux équations de ce modèle, un programme d'ordinateur a été écrit. C'est le but de ce document que de donner une description de ce programme d'ordinateur et de fournir la méthode pour l'utiliser.

UNCLASSIFIED

DEFENCE RESEARCH ESTABLISHMENT SUFFIELD
RALSTON ALBERTA

SUFFIELD TECHNICAL NOTE NO. 310

A COMPUTER PROGRAM FOR THE CALCULATION OF LIQUID OR VAPOUR PENETRATION THROUGH BARE AND CLOTHED SKIN (U)

by

W.R. McPherson

INTRODUCTION

A mathematical model (1) has been developed by Monaghan at DRES to predict the penetration of liquid or vapour through bare or clothed skin. The model, which has been used in toxicity studies, has been extensively evaluated against experimental data to establish its capabilities of predicting this penetration. This validation has shown that the model will provide a reasonably good description of the diffusion process that is taking place. According to the mathematics of the model, the derivation of estimates of mass penetration with time involves the solution of a partial differential equation. A computer program was written to calculate an approximate solution to the differential equation and to set up the various boundary conditions of the simulations which the model can take into account.

It is the objective of this paper to describe the computer program and to provide an operating manual for users of this manual. An elementary knowledge of the model and its terminology is required in order to use this program.

FEATURES OF THE MODEL

The penetration model was developed in order to better understand the hazards posed by toxic chemicals, either in the liquid or vapour phase, to men. In other words, this study dealt with an attempt to describe mathematically the physical processes occurring, with time, when liquid droplets impact on the clothing or bare skin of a man, or when a person either fully clothed or having bare skin is exposed to toxic chemical vapours. The passage of the contaminant through skin and clothing to the blood stream is a time dependent diffusion process.

The model is divided into five submodels. These submodels are summarized as the challenge of:

a) Vapour to bare skin

b) Liquid on bare skin

c) Vapour to a person wearing clothing

d) Liquid on a clothed person

e) Liquid pressed through clothing

The last challenge situation involves liquid being pressed through clothing layers. For example, when a person is contaminated by a chemical in the liquid phase, it impacts on the surface of his garments. If, however, he were now to force this liquid into his clothing by applying pressure, e.g., by sitting, then a more hazardous situation arises.

In addition to the above, two other provisions are part of the model. If the challenge is to a clothing ensemble, then provision is made for the removal of this contaminated clothing at some point in time. This is accomplished by zeroing the mass that exists in each of the clothing layers at that time. Similarily, the model will simulate the decontamination of skin to remove the surface contaminant.

The model can describe two types of skin situations. It is able to simulate real skin (human or animal, depending upon the values of the input parameters) or materials such as rubber dam. Rubber dam has been used by many workers in place of resected skin in laboratory penetration studies. The difference mathematically between the two is that it is assumed in the model that real skin can be divided into three distinct regimes, the horny surface layer, the transitional layer of the epidermis, and the blood stream (a sink), while a material such as rubber dam is assumed to be homogeneous.

CALCULATION OF PENETRATION

The object of this report is not to describe the mathematics of this model since these have been documented elsewhere. We wish only to give a very generalized outline of the approach taken to derive estimates. Penetration is described by a one-dimensional diffusion equation in a multi-layer system, i.e., skin and clothing, where absorptive and diffusive properties may vary from one layer to another and mass absorbed may decay with time in the system as a result of decomposition.

A simple implicit difference method was used to solve the diffusion equation. If we denote the vector of mass/unit area of the system by $\vec{M}_j$ (the components of $\vec{M}_j$ being the mass/unit area in each slice of the system at a given point in time equal to $j\Delta t$) and $\vec{M}_{j+1}$ is the vector of mass/unit area at the next increment in time, then the matrix equation describing the system can be represented by

$$\tilde{A}\vec{M}_{j+1} = \vec{M}_j+\vec{G}$$

$\tilde{A}$ is the matrix of coefficients of the absorptive and diffusive properties of the system and $\vec{G}$ is the vector of the boundary conditions. Since straightforward difference equations are used in the solution of the problem, $\tilde{A}$ is a tridiagonal matrix and its inverse can be readily calculated by the Gauss elimination method. Therefore, the solution to the problem is

$$\vec{M}_{j+1} = \tilde{A}^{-1} (\vec{M}_j + \vec{G})$$

given a set of initial conditions.

While the basic mathematics of the model appear quite simple, the elegance of it is attained through the control of the various boundary conditions possible, and the description and derivation of the components of $\tilde{A}$.

STRUCTURE OF THE COMPUTER PROGRAM

The computer program developed for the penetration model was originally designed to be used on a computer with only 8K words of core storage and therefore was written with a large number of subroutines to meet this constraint. In addition, this approach proved to be a convenient way of describing the model because of the large number of cases which it can simulate. In all, there are twelve subroutines, none of which are very large; this allowed for the maximum utilization of the limited core space originally available. The function of each of these subroutines is given in Table 1.

The program is written in the basic Fortran IV language of the DRES 1130 Model 2C computer (now 16K words of core storage). The program, in its present form, requires approximately 9K of core storage. It will allow up to four layers of clothing to be simulated over skin, each layer being subdivided into at most six slices. The skin is usually divided into 7 slices. There is a provision within the model to derive estimates for simulated skin, i.e. such materials as rubber dam; the mathematics describing the penetration process in skin are quite different to those for a simulated skin.

Figure 1 contains the block diagram of the computer model illustrating its basic structure. Appendix A provides a listing of the computer model in its present form.

INPUT PARAMETERS TO THE MODEL

The following is a description of the data required for input to the model. All input is by means of card.

Card 1 - 4I3 (Card 1 through 4 read in main line program)

N, NC, NS, NR

N = Total no. of slices skin in divided into (usually N=7)

NC = 1 Simulation is skin only
2 Clothing over skin

NS = 1 Average skin
2 Thin skin

NR = 0 Skin
1 Simulated skin

Card 2 - 8F10.0

RTHIN, VTHIN, RAVEG, VAVEG, U, DECAY, DELT

RTHIN, RAVEG - Resistance of skin types (min/cm)

VTHIN, VAVEG - Capacitance of skin types (cm)

U - Windspeed (cm/sec)

DECAY - Decay factor (% x 10^{-2} per minute)

DELT - Time increment in minutes

Card 3 - 8F10.0

TOTAL, TN, TD, TS, TR

TOTAL - Total model run time (hours)

TN - Type out increment (hours)

TD - Time of decontamination (hours)

TS - Time of removal of vapour source (hours)

TR - Time of removal of clothing (hours)

Card 4 - 80A1

A(I), I=1, 80

Variable heading information (alphanumeric)

These are the cards required for every type of run made. Any of the time inputs that are not needed may be left blank.

The program is subdivided into five subprograms which are:

1. Vapour on skin
2. Liquid on skin
3. Vapour on clothing
4. Liquid on clothing
5. Liquid pressed through clothing

Immediately after the above described four cards, the cards for the required subprogram are stacked.

(a) Skin only

Card 1 - 3I3

NL

NL = 1 Vapour on skin
2 Liquid on skin

If NL = 1

Card 2 - 8F10.0

C

C = Vapour concentration mg/cm^3

If NL = 2

Card 2 - 8F10.0

MASS(1), SVC

MASS(1) = MASS on sking (mg/cm^2)

SVC = Saturated vapour concentration (mg/cm^3)

(b) Clothing over skin

Card 1 - 3I3

NCL

NCL = 1 Vapour on clothing
2 Liquid on clothing
3 Liquid pressed through clothing

Card 2 - 3I3

NUM

NUM = No. of layers of clothing. The program will accept any number of layers but only the contents of the innermost 4 will be typed out.

In the case of simulated skin the contents of six clothing layers can be typed out. For more than six layers of clothing, the dimension statements must be altered.

Card 3A - I3, 2F10.0

MS(I), RT(I), VT(I)

One card per clothing layer (starting with the innermost)
MS = No. of slices that the layer will be subdivided into (usually five)
RT = Total resistances of each clothing layer (min/cm)
VT = Total capacitance of each clothing layer (cm)
If NCL = 1

Card 4 - 8F10.0

C

C = Concentration (mg/cm^3)

If NCL = 2

Card 4 - 8F10.0

MASS(1), CS

MASS(1) = Mass in mg/cm^2

CS = Saturated vapour concentration (mg/cm^3)

If NCL = 3

Card 4 - 8F10.0

(CS, MASS(I), I=1, NUM)

CS = Saturated vapour concentration (mg/cm^3)

MASS(I) = Mass in each layer at T=0 starting with the innermost (mg/cm^2)

Consecutive runs may be stacked one behind the other; a blank card will terminate the set of runs.

TABLE 1

SUBROUTINES OF THE PROGRAM CALLED "SKCLM"

SUBROUTINE NAME	PURPOSE
AGENT	Sets up resistance and capacitance coefficients i.e. the diffusive and absorptive properties, for the given skin type. Prints out heading and parameter information.
INDXX	Sets up run control indices.
CONST	Calculates the components of the matrix $\tilde{A}$, based on the input information.
CLOTH	Sets up boundary conditions when the model is used to simulate a clothing over skin ensemble. It can call the routines, VAPC, LIQC, LIQPC, CLREM.
VAPC	Boundary conditions for a vapour challenge against a cloth over skin situation.
LIQC	Boundary conditions for a liquid challenge against a cloth over skin situation.
LIQPC	Boundary conditions for liquid pressed into cloth in a cloth over skin situation.
SKIN	Sets up boundary conditions when model is used to simulate a vapour or liquid challenge against bare skin.
DECON	Simulates the removal or decontamination of penetrated mass from the skin.
TRID	Inverts the matrix $\tilde{A}$.
PROUT	Used to control printout of the mass penetrated with time.
SKCLM	The main line program.

REFERENCE

1. J. Monaghan, Private Communication.

APPENDIX A

COMPUTER PROGRAM LISTING

Subroutine - AGENT

```
      SUBROUTINE AGENT(RTHIN,VTHIN,RAVEG,VAVEG,U,R,V,RA1,RA2,N,NS,NC,NR,
     1DECAY,INDEX)
C
      DIMENSION R(2),V(2),INDEX(1),A(80)
      WRITE(3,1)
      READ(2,2) (A(I),I=1,80)
C     N=NO OF SLICES IN THE SKIN SYSTEM,NC=U NO CLOTHING,=2 CLOTHING
C     SYSTEM ADDED, NS=1 AVERAGE SKIN,=2 THIN SKIN, NR GREATER THAN ZERO
C     IMPLIES SIMULATED SKIN SYSTEM
C
      GO TO (15,16),NC
   15 IF(NR) 17,17,18
   17 WRITE(3,3)
      GO TO 21
   18 WRITE(3,4)
      GO TO 21
   16 IF(NR) 19,19,20
   19 WRITE(3,5)
      GO TO 21
   20 WRITE(3,6)
   21 WRITE(3,7) (A(I),I=1,80)
      IF(NR) 22,22,23
   22 GO TO (24,25),NS
   24 R(2)=RAVEG/FLOAT(N-2)
      V(2)=VAVEG/FLOAT(N-2)
      WRITE(3,8) RAVEG,VAVEG
      GO TO 26
   25 R(2)=RTHIN/FLOAT(N-2)
      V(2)=VTHIN/FLOAT(N-2)
      WRITE(3,9) RTHIN,VTHIN
      GO TO 26
   23 R(2)=RAVEG/FLOAT(N)
      V(2)=VAVEG/FLOAT(N)
      WRITE(3,10) RAVEG,VAVEG
   26 RA1=0.9/U**0.78
      RA2=RA1
      WRITE(3,11) U,RA1,DECAY
      DO 27 I=1,4
   27 INDEX(I)=0
C
    1 FORMAT(1H1,35X,'A MATHEMATICAL MODEL TO DESCRIBE AGENT PENETRATION
     1')
    2 FORMAT(80A1)
    3 FORMAT(51X,'THROUGH BARE SKIN')
    4 FORMAT(49X,'THROUGH SIMULATED SKIN')
    5 FORMAT(47X,'THROUGH CLOTHING AND SKIN')
    6 FORMAT(42X,'THROUGH CLOTHING AND SIMULATED SKIN')
    7 FORMAT(/,20X,80A1,/)
    8 FORMAT(54X,'AVERAGE SKIN',//,42X,'TOTAL RESISTANCE = ',F9.4,3X,'(M
     1IN/CM)',/,42X,'TOTAL CAPACITANCE = ',F9.2,2X,'(CM)')
    9 FORMAT(55X,'THIN SKIN',//,42X,'TOTAL RESISTANCE = ',F9.4,3X,'(MIN/
     1CM)',/,42X,'TOTAL CAPACITANCE = ',F9.2,2X,'(CM)')
   10 FORMAT(53X,'SIMULATED SKIN',//,42X,'TOTAL RESISTANCE = ',F9.4,3X,'
     1(MIN/CM)',/,42X,'TOTAL CAPACITANCE = ',F9.2,2X,'(CM)')
   11 FORMAT(/,42X,'WINDSPEED = ',F6.2,6X,'(CM/SEC)',/,42X,'BOUNDRY LAYE
     1R RESISTANCE',F9.3,2X,'(MIN/CM)',/,42X,'DECOMPOSITION RATE',10X,F5
     2.3,2X,'(1/MIN)')
C
      RETURN
      END
```

Subroutine - CØNST

```
      SUBROUTINE CONST(N,NC,NN,NUM,NK,DELT,DECAY,RA1,RA2,R,V,RT,VT,MS,A,
     1B,C,NR)
C
      DIMENSION V(2),R(2),VT(1),RT(1),MS(1),A(2),B(1),C(1)
C
      IF(NR) 1,1,2
    1 R(1)=R(2)/2.0
      R(N-1)=R(1)
      V(N)=5.*V(2)
      NK=N-1
      DO 10 I=3,NK
      V(I)=V(2)
      IF(I-(N-1)) 11,10,10
   11 R(I)=R(2)
   10 CONTINUE
      R(N)=RA1
      NK=N
      GO TO 3
    2 NK=N-1
      DO 4 I=1,NK
    4 R(I)=R(2)
      DO 5 I=1,N
    5 V(I)=V(2)
      NK=N
C
    3 GO TO (20,21),NC
C
   21 NK=N+1
      R(N)=RT(1)/(2.0*FLOAT(MS(1)))
      IF(NR) 6,6,7
    7 R(N)=R(N) + R(N-1)/2.0
    6 DO 12 J=1,NUM
      M1=MS(J) + NK-2
      DO 13 I=NK,M1
      R(I)=RT(J)/FLOAT(MS(J))
   13 V(I)=VT(J)/FLOAT(MS(J))
      V(M1+1)=VT(J)/FLOAT(MS(J))
      IF(J-NUM) 14,15,15
   15 R(M1+1)=RT(J)/(2.*FLOAT(MS(J)))
      GO TO 12
   14 R(M1+1)=(RT(J)/FLOAT(MS(J)) + RT(J+1)/FLOAT(MS(J+1)))/2.0
   12 NK=M1 + 2
      NN=M1 + 1
      R(NN) = R(NN) + RA2
      NK=NN
C
   20 B(1)=1.0
      A(2)=0.0
      DO 16 I=2,NK
      B(I)=1.0 + DELT*((R(I)+R(I-1))/(V(I)*R(I)*R(I-1)) + DECAY)
      C(I-1)= -DELT/(V(I)*R(I-1))
      IF(I-NK) 17,16,16
   17 A(I+1)= -DELT/(V(I)*R(I))
   16 CONTINUE
      IF(NR) 8,8,9
    9 B(1)=B(1) + DELT*(1.0/(V(1)*R(1)) + DECAY)
      A(2)=-DELT/(V(1)*R(1))
    8 CONTINUE
C
      RETURN
      END
```

Subroutine - INDXX

```
      SUBROUTINE INDXX(DELT,TOTAL,TN,TD,TS,TR,NTOTL,NTN,NTD,NTS,NTR)
C
      NTOTL = IFIX(TOTAL*60.0/DELT + 0.00001)
      NTN = IFIX(TN*60.0/DELT + 0.00001)
      NTD = IFIX(TD*60.0/DELT + 0.00001)
      NTS = IFIX(TS*60.0/DELT + 0.00001)
      NTR = IFIX(TR*60.0/DELT + 0.00001)
C
      WRITE(3,1)  TOTAL,TD,TN,DELT
    1 FORMAT(//,42X,'TIME CONSTANTS',/,44X,'TOTAL TIME',8X,'=',F7.2,3X,'
     1(HOURS)',/,44X,'TIME OF DECONTAM   =',F7.2,3X,'(HOURS)',/,44X,'TIME
     2 INCREMENT     =',F7.2,3X,'(HOURS)',/,44X,'DELT    =',F7.3,2X,'(MINU
     3TES)',///,53X,'TYPE OF SURFACE')
C
      RETURN
      END
```

Subroutine - CLØTH

```
      SUBROUTINE CLOTH(NCL,N,NN,NUM,NTIME,NTS,NTR,NK,INDEX,ITHLY,KTHSL,M
     1,R,B,G,U,DELT,RA2,DECAY,MASS,CS,MS,V,KIN)
C
      REAL M(70),MASS(6)
      DIMENSION INDEX(4),R(70),B(70),MS(6),V(70)
C
      GO TO (10,11,12),NCL
C
   10 IF(INDEX(1)) 13,13,14
   13 CALL VAPC(G,R,M,INDEX,DELT,NN)
   14 IF(NTS-NTIME) 17,16,15
   16 TIME=FLOAT(NTIME)*DELT/60.0
      WRITE(3,1) TIME
      G=0.0
      GO TO 17
C
   11 IF(INDEX(1)) 18,18,19
   18 CALL LIQC(U,R,V,B,M,RA2,DELT,DECAY,INDEX,MASS,CS,NN,G)
C
   19 IF(INDEX(2)) 20,20,21
   20 U=U + DELT*(CS*R(NN)/((R(NN)-RA2)*RA2)-M(NN)/(V(NN)*(R(NN)-RA2)))
      IF(U - MASS(1)) 21,22,22
   22 TIME=FLOAT(NTIME)*DELT/60.0
      WRITE(3,2) TIME
      G=0.0
      B(NN)=1.0 + DELT*((R(NN)+R(NN-1))/(V(NN)*R(NN)*R(NN-1)) + DECAY)
      INDEX(2)=INDEX(2) + 1
   21 GO TO 17
C
   12 IF(INDEX(1)) 23,23,24
   23 CALL LIQPC(CS,NUM,N,MS,M,ITHLY,KTHSL,KIN,INDEX)
   24 IF(INDEX(2)) 25,25,26
   25 IF(M(KTHSL)-CS*V(KTHSL)) 27,27,26
   27 KTHSL=KTHSL+1
      IF(KTHSL-NN) 28,28,29
   29 KTHSL=KTHSL-1
```

```
      NK=NN
      G=0.0
      INDEX(2)=INDEX(2) + 1
      GO TO 26
   28 GO TO 25
   26 IF((KTHSL-KIN)-MS(ITHLY)) 30,31,31
   31 TIME=FLOAT(NTIME)*DELT/60.0
      WRITE(3,3) ITHLY,TIME
      KIN=KIN+ MS(ITHLY)
      ITHLY=ITHLY + 1
      GO TO 26
   30 IF(INDEX(2)) 32,32,17
   32 NK=KTHSL-1
      G=DELT*CS/R(NK)
      IF(NK-(NN-2)) 33,35,35
   35 IF(NK-(NN-1)) 36,37,37
   36 M(NN-1)=M(NN-1)-DELT*(CS*(1./R(NN-2)+DECAY*V(NN-1))-M(NN-2)/(V(NN-
     12)*R(NN-2)))
      GO TO 37
   33 M(NK+1)=M(NK+1)-DELT*(CS*(1./R(NK) + DECAY*V(NK+1)) - M(NK)/(V(NK)
     1*R(NK)))
      L=NK+3
      DO 38 I=L,NN
   38 M(I-1)=M(I-1) - DELT*DECAY*CS*V(I-1)
   37 M(NN)=M(NN) - DELT*CS*(1./R(NN) + DECAY*V(NN))
   17 IF(NTR-NTIME) 15,34,15
   34 CALL CLREM(NTIME,N,NN,M,DELT,NK)
C
    1 FORMAT(/,42X,'SOURCE REMOVED A T=',F6.2,2X,'(HOURS)',/)
    2 FORMAT(/,42X,'LIQUID GONE AT T=',F6.2,2X,'(HOURS)',/)
    3 FORMAT(/,42X,'LIQUID ALL GONE IN LAYER  ',I2,' AT T=',F6.2,2X,'(HO
     1URS)',/)
C
   15 RETURN
      END
```

Subroutine - VAPC

```
      SUBROUTINE VAPC(G,R,M,INDEX,DELT,NN)
C
      REAL M(1)
      DIMENSION R(1),INDEX(1)
C
      READ(2,1) C
      WRITE(3,2) C
      G=(C/R(NN))*DELT
      DO 5 I=1,NN
    5 M(I)=0.0
      INDEX(1)=INDEX(1) + 1
    1 FORMAT(F10.0)
    2 FORMAT(/,42X,'VAPOUR ON CLOTHING, CONCENTRATION =',E11.4,' (MG/CM*
     1*3)')
C
      RETURN
      END
```

<u>Subroutine</u> - LIQC

```
      SUBROUTINE LIQC(U,R,V,B,M,RA2,DELT,DECAY,INDEX,MASS,CS,NN,G)
C
      REAL MASS(1),M(1)
      DIMENSION R(1),B(1),V(1),INDEX(1)
C
      READ(2,1) MASS(1),CS
      WRITE(3,2) MASS(1),CS
      U=0.0
      G=(CS/(R(NN)-RA2))*DELT
      B(NN)=1.+DELT*((R(NN)-RA2+R(NN-1))/(V(NN)*(R(NN)-RA2)*R(NN-1)) +
     1 DECAY)
      DO 3 I=1,NN
    3 M(I)=0.0
      INDEX(1)=INDEX(1) + 1
    1 FORMAT(2F10.0)
    2 FORMAT(/,42X,'LIQUID ON CLOTHING MASS =',F7.3,2X,'(MG/CM**2)',/,5
     18X,'SVC =',E11.4,2X,'(MG/CM**3)')
C
      RETURN
      END
```

<u>Subroutine</u> - LIQPC

```
      SUBROUTINE LIQPC(CS,NUM,N,MS,M,ITHLY,KTHSL,KIN,INDEX)
C
      REAL M(1),MASS(6)
      DIMENSION MS(1),INDEX(1)
C
      READ(2,1) CS,(MASS(I),I=1,NUM)
      WRITE(3,2) CS,(I,MASS(I),I=1,NUM)
      DO 30 I=1,NUM
      IF(MASS(I)) 30,30,31
   30 CONTINUE
   31 ITHLY=I
      KTHSL=N+1
      IF(ITHLY-1) 32,32,33
   33 DO 34 I=2,ITHLY
   34 KTHSL=KTHSL + MS(I-1)
   32 KIN=KTHSL
      DO 35 I=2,KTHSL
   35 M(I-1)=0.0
      NN=KTHSL
      DO 36 J=ITHLY,NUM
      NK=MS(J)+NN-1
      DO 37 L=NN,NK
   37 M(L)=MASS(J)/FLOAT(MS(J))
   36 NN=NK+1
      INDEX(1)=INDEX(1) + 1
C
    1 FORMAT(8F10.0)
    2 FORMAT(/,42X,'LIQUID PRESSED THROUGH CLOTHING',/,58X,'SVC =',E11.4
     1,2X,'(MG/CM**3)',//,(63X,I1,7X,F7.3,/))
      RETURN
      END
```

Subroutine - CLREM

```
      SUBROUTINE CLREM(NTIME,N,NN,M,DELT,NK)
      REAL M(1)
      K=N+1
      TIME=FLOAT(NTIME)*DELT/60.0
      WRITE(3,1) TIME
      DO 2 I=K,NN
    2 M(I)=0.0
      NK=N
    1 FORMAT(/,42X,'CLOTHING REMOVED AT T=',F6.2,2X,'(HOURS)',/)
C
      RETURN
      END
```

Subroutine - SKIN

```
      SUBROUTINE SKIN(NL,N,NTS,NTIME,INDEX,DELT,M,R,V,B,G,DECAY,KN,MASS)
C
      REAL M(1),MASS(1),KN
      DIMENSION INDEX(2),R(1),V(1),B(1)
C
      GO TO (10,11),NL
C
   10 IF(INDEX(1)) 12,12,13
   12 READ(2,1) C
      WRITE(3,2) C
      G=(C/R(N))*DELT
      DO 30 I=1,N
   30 M(I)=0.0
      INDEX(1)=INDEX(1) + 1
   13 IF(NTS-NTIME) 14,15,14
   15 TIME=FLOAT(NTIME)*DELT/60.0
      WRITE(3,3) TIME
      G=0.0
      GO TO 14
C
   11 IF(INDEX(1)) 16,16,17
   16 READ(2,1) MASS(1),SVC
      WRITE(3,4) MASS(1),SVC
      DO 31 I=2,N
   31 M(I-1)=0.0
      M(N)=MASS(1)
      IF(MASS(1)/V(N)-SVC) 18,18,19
   18 WRITE(3,5)
      G=0.0
      INDEX(1)=INDEX(1) + 1
      GO TO 14
C
   19 KN=V(N)*SVC/MASS(1)
      WRITE(3,6)
      G=-(DELT*KN*MASS(1)/(V(N)*R(N)))
      B(N)=(1.0 + DELT*(1./(V(N)*R(N-1)) + DECAY))
      INDEX(1)=INDEX(1) + 1
      INDEX(2)=INDEX(2) + 1
```

```
C
   17 IF(INDEX(2)-1) 14,20,14
   20 IF(M(N)-KN*MASS(1)) 21,21,14
   21 G=0.0
      B(N)=1. + DELT*((R(N)+R(N-1))/(V(N)*R(N)*R(N-1)) + DECAY)
      INDEX(2)=INDEX(2) + 1
C
   14 CONTINUE
    1 FORMAT(2F10.0)
    2 FORMAT(/,42X,'VAPOUR ON SKIN, CONCENTRATION =',E11.4,' (MG/CM**3)'
     1)
    3 FORMAT(/,42X,'SOURCE REMOVED AT T=',F6.2,2X,'(HOURS)',/)
    4 FORMAT(/,42X,'LIQUID ON SKIN MASS=',F7.3,2X,'(MG/CM**2)',/,56X,' S
     1VC =',F7.3,2X,'(MG/CM**3)')
    5 FORMAT(/,42X,'THE DENSITY APPLIED IS LESS THAN THE SVC',/)
    6 FORMAT(/,42X,'THE DENSITY APPLIED IS GREATER THAN THE SVC',/)
C
      RETURN
      END
```

Subroutine - DECØN

```
      SUBROUTINE  DECON(NTIME,A,B,C,N,G,DELT,M)
      REAL M(1)
      DIMENSION A(1),B(1),C(1)
      TIME=FLOAT(NTIME)*DELT/60.0
      WRITE(3,1) TIME
      C(N-1)=0.0
      B(N)=0.0
      A(N)=0.0
      G=0.0
      M(N)=0.0
    1 FORMAT /,42X, DECONTAMINATION AT T  ,F6.2,2X,  HOURS  ,/
C
      RETURN
      END
```

Subroutine - TRID

```
      SUBROUTINE TRID(M,A,B,C,NK,GG)
C
      REAL M(1)
      DIMENSION A(1),B(1),C(1),Q(70),W(70),G(70)
C
      M(NK)=M(NK)+GG
      W(1)=B(1)
      G(1)=M(1)/W(1)
      DO 1 I=2,NK
      Q(I-1)=C(I-1)/W(I-1)
      W(I)=B(I)-A(I)*Q(I-1)
    1 G(I)=(M(I)-A(I)*G(I-1))/W(I)
C
      M(NK)=G(NK)
      DO 2 I=2,NK
      K=NK+1-I
    2 M(K)=G(K)-Q(K)*M(K+1)
C
      RETURN
      END
```

Subroutine - PRØUT

```
      SUBROUTINE PROUT(NC,NR,NCL,NUM,N,NTIME,M,MS,U,DELT,INDEX)
C
      REAL M(1)
      DIMENSION INDEX(3),SUM(7),MS(1)
C
      TIME=FLOAT(NTIME)*DELT/60.0 + 0.00001
      DO 20 I=1,7
   20 SUM(I)=0.0
C
      IF(INDEX(3)) 21,21,22
   21 WRITE(3,1)
      INDEX(3)=INDEX(3) + 1
      IF(NR) 23,23,24
   23 GO TO (25,26),NC
   25 WRITE(3,2)
      GO TO 22
   26 IF(NUM-4) 27,27,28
   27 NM=NUM
      GO TO 29
   28 NM=4
   29 GO TO (30,31,32,33),NM
   30 WRITE(3,3)
      GO TO 22
   31 WRITE(3,4)
      GO TO 22
   32 WRITE(3,5)
      GO TO 22
   33 WRITE(3,6)
      GO TO 22
C
   24 GO TO (34,35),NC
   34 WRITE(3,7)
      GO TO 22
   35 IF(NUM-6) 36,36,37
   36 NM=NUM
      GO TO 38
   37 NM=6
   38 GO TO (39,40,41,42,43,44),NM
   39 WRITE(3,8)
      GO TO 22
   40 WRITE(3,9)
      GO TO 22
   41 WRITE(3,10)
      GO TO 22
   42 WRITE(3,11)
      GO TO 22
   43 WRITE(3,12)
      GO TO 22
   44 WRITE(3,13)
C
   22 IF(NR) 45,45,46
   45 DO 47 I=3,N
   47 SUM(1)=SUM(1) + M(I-1)
```

```
      GO TO 48
   46 DO 49 I=1,N
   49 SUM(1)=SUM(1) + M(I)
   48 GO TO (50,51),NC
   50 IF(NR) 52,52,53
   52 WRITE(3,14) TIME,M(1),SUM(1),M(N)
      GO TO 54
   53 WRITE(3,15) TIME,SUM(1)
      GO TO 54
   51 K=N+1
      KNK=NM + 1
      DO 55 J=2,KNK
      KK=K-1+MS(J-1)
      DO 56 I=K,KK
   56 SUM(J)=SUM(J) + M(I)
   55 K=KK+1
      IF(NR) 57,57,58
   57 GO TO (59,60,59),NCL
   59 WRITE(3,16) TIME,M(1),SUM(1),M(N),(SUM(I),I=2,KNK)
      GO TO 54
   60 WRITE(3,17) TIME,U,M(1),SUM(1),M(N),(SUM(I),I=2,KNK)
      GO TO 54
   58 GO TO (61,62,61),NCL
   61 WRITE(3,18) TIME,(SUM(I),I=1,KNK)
      GO TO 54
   62 WRITE(3,19) TIME,U,(SUM(I),I=1,KNK)
    1 FORMAT('1',44X,'MASS IN EACH LAYER IN MGMS/CM**2',//)
    2 FORMAT(14X,'TIME',15X,'SYSTEMIC',2(15X,'MASS REMAINING IN'),/,13X,
     1'HOURS',17X,'DOSE',17X,'TRANSITIONAL LAYER',17X,'HORNY LAYER',/)
    3 FORMAT(2X,'TIME',4X,'U',6X,'SYSTEMIC',9X,'MASS IN',8X,'MASS IN',8X
     1,'MASS IN',/,1X,'HOURS',1X,'MG/CM**2',4X,'DOSE',9X,'TRANSITIONAL',
     26X,'HORNY',7X,'FIRST CLOTH',/,35X,3('LAYER',10X),/)
    4 FORMAT(2X,'TIME',4X,'U',6X,'SYSTEMIC',9X,'MASS IN',8X,'MASS IN',8X
     1,'MASS IN',8X,'MASS IN',/,1X,'HOURS',1X,'MG/CM**2',4X,'DOSE',9X,'T
     2RANSITIONAL',6X,'HORNY',7X,'FIRST CLOTH',4X,'SECOND CLOTH',/,35X,4
     3('LAYER',10X),/)
    5 FORMAT(2X,'TIME',4X,'U',6X,'SYSTEMIC',9X,'MASS IN',8X,'MASS IN',8X
     1,'MASS IN',8X,'MASS IN',8X,'MASS IN',/,1X,'HOURS',1X,'MG/CM**2',4X
     2,'DOSE',9X,'TRANSITIONAL',6X,'HORNY',7X,'FIRST CLOTH',4X,'SECOND C
     3LOTH',3X,'THIRD CLOTH',/,35X,5('LAYER',10X),/)
    6 FORMAT(2X,'TIME',4X,'U',6X,'SYSTEMIC',9X,'MASS IN',8X,'MASS IN',8X
     1'MASS IN',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',/,1X,'HOURS',1X,'
     2MG/CM**2',4X,'DOSE',9X,'TRANSITIONAL',6X,'HORNY',7X,'FIRST CLOTH',
     34X,'SECOND CLOTH',3X,'THIRD CLOTH',4X,'FOURTH CLOTH',/,35X,5('LAYE
     4R',10X),'LAYER',/)
    7 FORMAT(50X,'TIME',10X,'MASS IN',/,49X,'HOURS',9X,'SIMULATED',/,65X
     2,'SKIN',/)
    8 FORMAT(2X,'TIME',4X,'U',8X,'MASS IN',8X,'MASS IN',/,' HOURS MG/CM*
     1*2    SIMULATED',5X,'FIRST CLOTH',/,20X,'SKIN',11X,'LAYER',/)
    9 FORMAT(2X,'TIME',4X,'U',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',/,'
     1 HOURS MG/CM**2    SIMULATED',5X,'FIRST CLOTH',4X,'SECOND CLOTH',/,
     220X,'SKIN',11X,'LAYER',10X,'LAYER',/)
   10 FORMAT(2X,'TIME',4X,'U',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',8X,
     1'MASS IN',/,' HOURS MG/CM**2    SIMULATED',5X,'FIRST CLOTH',4X,'SEC
     2OND',3X,'THIRD CLOTH',/,20X,'SKIN',11X,3('LAYER',10X),/)
   11 FORMAT(2X,'TIME',4X,'U',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',8X,
     1'MASS IN',8X,'MASS IN',/,' HOURS MG/CM**2    SIMULATED',5X,'FIRST C
     2LOTH    SECOND CLOTH    THIRD CLOTH     FOURTH CLOTH',/,20X,'SKIN',1
     31X,4('LAYER',10X),/)
```

```
   12 FORMAT(2X,'TIME',4X,'U',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',8X,
     1'MASS IN',8X,'MASS IN',8X,'MASS IN',/,' HOURS MG/CM**2   SIMULATED
     2',5X,'FIRST CLOTH    SECOND CLOTH    THIRD CLOTH     FOURTH CLOTH
     3FIFTH CLOTH',/,20X,'SKIN',11X,5('LAYER',10X),/)
   13 FORMAT(2X,'TIME',4X,'U',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',8X,
     1'MASS IN',8X,'MASS IN',8X,'MASS IN',8X,'MASS IN',/,' HOURS MG/CM**
     22    SIMULATED',5X,'FIRST CLOTH    SECOND CLOTH    THIRD CLOTH     FO
     3URTH CLOTH    FIFTH CLOTH    SIXTH CLOTH',/,20X,'SKIN',11X,5('LAYER
     4',10X),'LAYER',/)
   14 FORMAT(12X,F7.3,13X,F10.7,18X,F10.7,21X,F10.7)
   15 FORMAT(48X,F7.3,8X,F9.6)
   16 FORMAT(F7.3, 8X,7(2X,F10.7,3X))
   17 FORMAT(F7.3,F8.5,7(2X,F10.7,3X))
   18 FORMAT(F7.3, 8X,7(2X,F10.7,3X))
   19 FORMAT(F7.3,F8.5,7(2X,F10.7,3X))
C
   54 RETURN
      END
```

Main line program named - SKCLM

```
      REAL M(70),MASS(6),KN
      DIMENSION A(70),B(70),C(70),R(70),V(70),RT(6),VT(6),MS(6),INDEX(4)
C
   21 READ(2,1) N,NC,NS,NR
      IF(N) 25,25,26
   26 READ(2,2) RTHIN,VTHIN,RAVEG,VAVEG,U,DECAY,DELT
      READ(2,2) TOTAL,TN,TD,TS,TR
      CALL AGENT(RTHIN,VTHIN,RAVEG,VAVEG,U,R,V,RA1,RA2,N,NS,NC,NR,DECAY,
     1INDEX)
C
      GO TO (10,11),NC
C
   10 READ(2,3) NL
      GO TO 12
C
   11 READ(2,3) NCL
      READ(2,4) NUM,(MS(I),RT(I),VT(I),I=1,NUM)
      WRITE(3,5)
      WRITE(3,6) (I,MS(I),RT(I),VT(I),I=1,NUM)
C
   12 CALL INDXX(DELT,TOTAL,TN,TD,TS,TR,NTOTL,NTN,NTD,NTS,NTR)
      CALL CONST(N,NC,NN,NUM,NK,DELT,DECAY,RA1,RA2,R,V,RT,VT,MS,A,B,C,NR
     1)
      NTT=NTN
C
      NTIME = 1
C
   22 GO TO (13,14),NC
C
   13 CALL SKIN(NL,N,NTS,NTIME,INDEX,DELT,M,R,V,B,G,DECAY,KN,MASS)
      GO TO 15
C
   14 CALL CLOTH(NCL,N,NN,NUM,NTIME,NTS,NTR,NK,INDEX,ITHLY,KTHSL,M,R,B,G
     1,U,DELT,RA2,DECAY,MASS,CS,MS,V,KIN)
C
   15 IF(NTIME-NTD) 16,17,16
   17 CALL DECON(NTIME,A,B,C,N,G,DELT,M)
C
   16 CALL TRID(M,A,B,C,NK,G)
      IF(NTIME-NTT) 18,19,19
   19 CALL PROUT(NC,NR,NCL,NUM,N,NTIME,M,MS,U,DELT,INDEX)
      NTT=NTN + NTIME
C
   18 IF(NTIME - NTOTL) 20,21,21
   20 NTIME=NTIME + 1
      GO TO 22
C
    1 FORMAT(4I3)
    2 FORMAT(8F10.0)
    3 FORMAT(I3)
    4 FORMAT(I3,/,(I3,F10.0,F10.0))
    5 FORMAT(//,42X,'CLOTHING PARAMETERS',/,44X,'LAYER',2X,'MS',2X,'RESI
     1STANCE',2X,'CAPACITANCE',/,56X,'(MIN/CM)',6X,'(CM)',/)
    6 FORMAT(46X,I1,4X,I2,3X,F6.3,6X,F8.2)
C
   25 CALL EXIT
      END
```



FIGURE 1: BLOCK DIAGRAM OF THE COMPUTER MODEL NAMED "SKCLM"

412 42d

0201 DRES
0202 403104

UNCLASSIFIED
Security Classification

DOCUMENT CONTROL DATA – R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall document is classified)

1. ORIGINATING ACTIVITY 0204a Defence Research Establishment Suffield, 0204b Ralston ALTA (CAN)	2a. DOCUMENT SECURITY CLASSIFICATION UNCLASSIFIED 2b. GROUP

3. DOCUMENT TITLE
0?40 A Computer Program for the Calculation of Liquid or Vapour Penetration Through Bare and Clothed Skin (U)

4. DESCRIPTIVE NOTES (Type of report and inclusive dates) Technical Note

5. AUTHOR(S) (Last name, first name, middle initial)
1101
McPherson, W.R.

6 DOCUMENT DATE 46 April 1972	7a. TOTAL NO. OF PAGES 0901 21 23	7b. NO. OF REFS 0902 4 0

8a. PROJECT OR GRANT NO. 35 D - 20-20-32	9a ORIGINATOR'S DOCUMENT NUMBER(S) 0203 Suffield Technical Note No. -310
8b. CONTRACT NO.	9b. OTHER DOCUMENT NO.(S) (Any other numbers that may be assigned this document)

10. DISTRIBUTION STATEMENT
"Qualified requesters may obtain copies of this document from their defence documentation centre."

11. SUPPLEMENTARY NOTES	12. SPONSORING ACTIVITY

13. ABSTRACT

A mathematical model has been developed by Monaghan at DRES to predict the penetration of liquid or vapour through bare or clothed skin; in order to calculate solutions to the equations of this model, a computer program was written. It is the objective of this paper to give a description of this computer program and provide the method of using it.

DSIS
70-070

DC

UNCLASSIFIED
Security Classification

KEY WORDS

1. Predictive Model
2. Computer Program
3. Liquid Diffusion
4. Vapour Diffusion
5. Clothing Penetration
6. Skin Absorption

72-03509
#14540

INSTRUCTIONS

1. ORIGINATING ACTIVITY. Enter the name and address of the organization issuing the document.

2a. DOCUMENT SECURITY CLASSIFICATION: Enter the overall security classification of the document including special warning terms whenever applicable.

2b. GROUP: Enter security reclassification group number. The three groups are defined in Appendix 'M' of the DRB Security Regulations.

3. DOCUMENT TITLE: Enter the complete document title in all capital letters. Titles in all cases should be unclassified. If a sufficiently descriptive title cannot be selected without classification, show title classification with the usual one-capital-letter abbreviation in parentheses immediately following the title.

4. DESCRIPTIVE NOTES: Enter the category of document, e.g. technical report, technical note or technical letter. If appropriate, enter the type of document, e.g. interim, progress, summary, annual or final. Give the inclusive dates when a specific reporting period is covered.

5. AUTHOR(S): Enter the name(s) of author(s) as shown on or in the document. Enter last name, first name, middle initial. If military, show rank. The name of the principal author is an absolute minimum requirement.

6. DOCUMENT DATE: Enter the date (month, year) of Establishment approval for publication of the document.

7a. TOTAL NUMBER OF PAGES: The total page count should follow normal pagination procedures, i.e., enter the number of pages containing information.

7b. NUMBER OF REFERENCES: Enter the total number of references cited in the document.

8a. PROJECT OR GRANT NUMBER If appropriate, enter the applicable research and development project or grant number under which the document was written.

8b. CONTRACT NUMBER: If appropriate, enter the applicable number under which the document was written.

9a. ORIGINATOR'S DOCUMENT NUMBER(S): Enter the official document number by which the document will be identified and controlled by the originating activity. This number must be unique to this document.

9b. OTHER DOCUMENT NUMBER(S). If the document has been assigned any other document numbers (either by the originator or by the sponsor), also enter this number(s).

10. DISTRIBUTION STATEMENT Enter any limitations on further dissemination of the document, other than those imposed by security classification, using standard statements such as.

(1) "Qualified requesters may obtain copies of this document from their defence documentation center."

(2) "Announcement and dissemination of this document is not authorized without prior approval from originating activity."

11. SUPPLEMENTARY NOTES: Use for additional explanatory notes.

12. SPONSORING ACTIVITY: Enter the name of the departmental project office or laboratory sponsoring the research and development. Include address.

13. ABSTRACT Enter an abstract giving a brief and factual summary of the document, even though it may also appear elsewhere in the body of the document itself. It is highly desirable that the abstract of classified documents be unclassified. Each paragraph of the abstract shall end with an indication of the security classification of the information in the paragraph (unless the document itself is unclassified) represented as (TS), (S), (C), (R), or (U).

The length of the abstract should be limited to 20 single-spaced standard typewritten lines, 7½ inches long.

14. KEY WORDS: Key words are technically meaningful terms or short phrases that characterize a document and could be helpful in cataloging the document. Key words should be selected so that no security classification is required. Identifiers, such as equipment model designation, trade name, military project code name, geographic location, may be used as key words but will be followed by an indication of technical context.